ChatScript Debugging Manual

© **Bruce Wilcox, gowilcox@gmail.com**

Revision 11-16-2012 cs2.97

You've written script. It doesn't work. Now what? Now you need to debug it, fix it, and recompile it. Debugging is mostly a matter of tracing what the system does and finding out where it doesn't do what you expected. Debugging mostly done by issuing commands to the engine, as opposed to chatting.

If the system detects bugs during execution, they go into *TMP/bugs.txt* You can erase the entire contents of the TMP directory any time you want to. But odds are this is not your problem. Debugging generally requires use of some :xxxx commands. I don't always remember them all, so at times I might simply say
:help
to get a list of them and a rough description.

Before it goes wrong during execution

Before you chat with your chatbot and discover things don't work, there are a couple of things to do first, to warn you of problems.

Compiling (:build)

Of course, you started with :build to compile your script. It wouldn't have passed a script that was completely wrong, but it might have issued warnings. That's also not likely to be your problem, but let's look at what it might have told you as a warning. The system will warn you as it compiles and it will summarize its findings at the end of the compile. The compilation messages occur on screen and in the log file.

The most significant warnings are a reference to an undefined concept or an undefined ^reuse label. E.g.,

*** Warning- missing set definition ~car_names
*** Warning- Missing cross-topic label ~allergy.JOHN for reuse

Those you must fix because clearly they are wrong, although the script will execute fine everywhere but in those places.

*** Warning- flowglow is unknown as a word in pattern

Warnings about words in patterns that it doesn't recognize or it recognizes in lower case but you used upper case may or may not matter. Neither of these is wrong, unless you didn't intend it. Words it doesn't recognize arise either because you made a typo (requiring you fix it) or simply because the word isn't in the dictionary. Words in upper case are again words it knows as lower case, but you used it as upper case. Maybe right or wrong.

Editing the main dictionary is not a task for the faint-hearted. But ChatScript maintains secondary dictionaries in the TOPIC folder and those are easy to adjust. To alter them, you can define concepts that add local dictionary entries. The names of important word type bits are in src/dictionarySystem.h but the basics are NOUN, VERB, ADJECTIVE, and ADVERB.

```
concept: ~morenoun NOUN (fludge flwoa  boara)
```

A concept like the above, declared before use of those words, will define them into the local dictionary as nouns and suppress the warning message. You can be more specific as well with multiple flags like this:

```
concept: ~myverbs VERB VERB_INFINITIVE (spluat  babata)
```

You can define concepts at level 0 and/or level 1 of the build, so you can put define new words whenever you need to.

When build is complete, it will pick up where it left off with the user (his data files unchanged). If you want to automatically clear the user and start afresh, use
:build xxx reset

Verification (:verify)

When I write a topic, before every rejoinder and every responder, I put a sample input comment. This has #! as a prefix. E.g.,

```
topic: ~mytopic (keyword)
t: This is my topic.
        #! who cares
        a: (who)  I care.

#! do you love me
?: (do you love)  Yes.

#! I hate food
s: (I * ~hate * food) Too bad.
```

This serves two functions. First, it makes it easy to read a topic-- you don't have to interpret pattern code to see what is intended. Second, it allows the system to verify your code in various ways. It is the “unit test” of a rule. If you've annotated your topics in this way, you can issue

:verify
:verify ~topicname
:verify keyword
:verify pattern ~topic
:verify blocking

There are several verifications the system can do. The first command verifies all topics in all ways. The second restricts itself to the specific topic listed in all ways. The third verifies keywords of all topics. The fourth verifies just the patterns of rules in the given topic. The fifth verifies blocking of all topics.

:verify keyword: For responders, does the sample input have any keywords from the topic. If not, there may be an issue. Maybe the sample input has some obvious topic-related word in it, which should be added to the keywords list. Maybe it's a responder you only want matching if the user is in the current topic. E.g.,

```
#! Do you like swirl?
?: ( swirl) I love raspberry swirl
```

can match inside an ice cream topic but you don't want it to react to *Does her dress swirl?*.

Or maybe the sample input has no keywords but you do want it findable from outside (E.g., an idiom), so you have to make it happen. When a responder fails this test, you have to either add a keyword to the topic, revise the sample input, add an idiom table entry, or tell the system to ignore keyword testing on that input.

You can suppress keyword testing by augmenting the comment on the sample input with !K.

```
#!!K this is input that does not get keyword tested.
```

:verify pattern: For responders and rejoinders, the system takes the sample input and tests to see if the pattern of the following rule actually would match the given input. Failing to match means the rule is either not written correctly for what you want it to do or you wrote a bad sample input and need to change it.

Rules that need variables set certain ways can do variable assigns ($ or _) at the end of the comment. You can also have more than one verification input line before a rule.

```
#! I am male  $gender = male
#! I hate males $gender = male
s: ($gender=male I * male) You are not my type
```

If you want to suppress testing, add !P to the comment.

```
#!!P This doesn't get pattern testing.
```

If you want to suppress pattern and keyword testing, just use K and P in either order:

```
#!!KP this gets neither testing.
```

You can also test that the input does not match the pattern by using #!!R instead of #!, though unless you were writing engine diagnostic tests this would be worthless to you.

:verify blocking: Even if you can get into the topic from the outside and the pattern matches, perhaps some earlier rule in the topic can match and take control instead. This is

called blocking. One normally tries to place more specific responders and rejoinders before more general ones. The below illustrates blocking for *are your parents alive?* The sentence will match the first rule before it ever reaches the second one.

```
#! do you love your parents
?: ( << you parent >>) I love my parents       *** this rule triggers by mistake
#! are your parents alive
?: ( << you parent alive >>) They are still living
```

The above can be fixed by reordering, but sometimes the fix is to clarify the pattern of the earlier rule.

```
#! do you love your parents
?: ( ![alive living dead] << you parent >>) I love my parents
#! are your parents alive
?: ( << you parent alive >>) They are still living
```

Sometimes you intend blocking to happen and you just tell the system not to worry about it using !B.

```
#! do you enjoy salmon?
?: ( << you ~like salmon >>) I love salmon
#!!B do you relish salmon?
?: (<< you ~like salmon >>) I already told you I did.
```

The blocking test presumes you are within the topic and says nothing about whether the rule could be reached if you were outside the topic. That's the job of the keyword test. And it only looks at you sample input. Interpreting your pattern can be way too difficult.

If :trace has been set non-zero, then tracing will be turned off during verification, but any rules that fail pattern verification will be immediately be rerun with tracing on so you can see why it failed.

Passing verification means that each topic, in isolation, is plausibly scripted. Other topics might still grab control, but that's a different issue. So, you should verify first and live test second. But things can still go wrong. Or maybe you need to debug to find out why :verify pattern failed.

Keyword overlap

As you assign keywords to topics, at times you will probably get excessive. Some topics will share keywords with other topics. For some things, this is reasonable. “quark” is a find keyword for a topic on chees and one on astronomy. But odds are “family” is not a great keyword for a topic on money. Often an extraneous keyword won't really matter, but if the system is looking for a topic to gambit based on “family”, you really don't want it distracted by a faulty reference to money. That is, you want to know what keywords are shared across topics and then you can decide if that's appropriate.

Validation

Having built a functioning chatbot with lots of topics, I like to insure I don't damage old material in the future, so I write a list of test questions or statements for each topic. I refer to this as validation. I run the chatbot against this file using :source. I take the resulting log file, move it to LOGS directory and type :trim 2 . This generates a file TMP/tmp.txt which lists three things: the topic an answer came from, the sample input, and the actual output. I save this file away somewhere.

Later on, I can show the chatbot is undamaged by :source of the validation input file, doing a :trim 2 on the moved log file, and then doing a "diff" using a diff tool on the new output of trim vs the old output. If I see changes, I have to account for them either by fixing the chatbot, or replacing the standard output base with the new result.

:overlappingkeys
will show you a topic and all keywords (directly or inherited from concepts which are keywords) which exist in multiple topics. Then you can decide if the keywords for a topic make sense or might cause mischief. You can restrict it to looking at a single topic: :overlappingkeys ~mytopic

:topickeys (topic) will show you all keywords of a topic (or of all topics if you omit naming a topic).

After it goes wrong during execution

The most common issue is that ChatScript will munge with your input in various ways so you don't submit what you think you are submitting. The substitutions file will change words or phrases. Spell correction will change words. Proper name and number merging will adjust words. And individual words of yours will be merged into single words if WordNet lists them as a multiple-word (like "TV_star") . So you really need to see what your actual input ended up being before you can tell if your pattern was correct or not. Thus the most common debug command is :prepare.

:prepare this is my sample input

This shows you how the system will tokenize a sentence and what concepts it matches. It's what the system does to prepare to match topics against your input. From that you can deduce what patterns would match and what topics might be invoked.

If you give no argument to :prepare, it toggles a mode that will always just prepare every input thereafter without actually executing the bot on it (or not).

:pos

This is a subset of :prepare that just runs the POS-tagger parser on the input you supply. I use it to debug the system. It either is given a sentence or toggles a mode if not (just like :prepare). It also displays pronoun data gathered from the input.

:trace all
:trace none

The ultimate debugging command dumps a trace of everything that happens during execution onto screen and into the log file. After entering this, you type in your chat and watch what happens (which also gets dumped into the current log file). Problem is, it's potentially a large trace. You really want to be more focused in your endeavor.

:trace ~education **-** this enables tracing for this topic and all the topics it calls. Call it again to restore to normal.

:trace !~education **-** this disables current tracing for this topic and all the topics it calls when :trace all is running. Call it again to restore to normal.

:trace ~education.school – this traces all top-level rules in ~education that you named *school* (and its rejoinders and anything it calls. Call it again to turn off the trace.

You can insert a trace command in the data of a table declaration, to trace a table being built (the log file will be from the build in progress). E.g.,

table: ~capital (^base ^city)
_9 = join(^city , _ ^base)
^createfact(_9 member ~capital)
DATA:
:trace all
Utah Salt_lake_city
:trace none

You can insert a trace in the list of files to process for a build. From files1.txt

RAWDATA/simplecontrol.top # conversational control
:trace all
RAWDATA/simpletopic.top # things we can chat about
:trace none

And you can insert a trace in the list of commands of a topic file:

t*opic: foo [...]*
.....
:trace all

Tracing is also good in conjuction with some other commands that give you a restricted view.

Standard debugging practice

When I test my bots (assuming they pass verification), I chat with them until I get an answer I don't like. I then ask it why it generated the answer it did.

:why

This specifies the rules that generated the actual output and what topics they came from. I can often see, looking at the rule, why I wouldn't want it to match and go fix that rule. That doesn't address why some rule I want to match failed, so for that I'll need typically need tracing. So I enable some trace (being lazy I typically do :trace all) and then I say

:revert

:revert tells the system to rerun the most recent input on the state of the system a moment ago, and retry it. It should do exactly what it did before, but this time with tracing on. It performs this magic in stand-alone mode by copying the user's topic file into the TMP directory before each volley, so it can back up one time if needed. Because it operates from that tmp copy, you can, if your log file is currently messy, merely erase all the contents of the USER folder before executing the revert and the log trace will only be from this input.

:skip n

The system will disable the next n gambits of the current topic, and tell you where you will end up next. Thereafter your next simple input like “ok” will execute that gambit, and the n previous will already have been used up.

:retry

Similar to :revert, :retry backs up to try the prior context again, but instead of repeating the prior input, you can put in different input this time. It will prompt you for that input if none is given on the command, or if you say:

:retry what is your name

then it will use that as the replacement input.

:say

Sometimes you need the system to set up a situation (typically pronoun resolution). :say xxxx will tell the system to generate xxxx as its output and then postprocess it as if it had generated it on its own.

:set %xxx yyyy

You can explicitly set system variables to return a fixed value instead of what they normally would. Thus for testing you could set the %day or other system variables.

:show

The :show choices are all toggles. Each call flips its state.

:show number displays the current input number at the front of the bot's output. It is always shown in the log, but this shows it on-screen. I use this before running a large regression test like :source REGRESS/bigregress.txt so I will know how far we have gotten while it's running.

:show pos displays details on the POS-tagging parser process. Not useful for a user, but useful to me in debugging the engine itself. When you do :prepare or get a :trace, the results of POS tagging are shown after the individual words and their marked concepts.

:show topic displays the current topic the bot is in, prefixed to its output.

:show why this turns on :why for each volley so you always see the rule causing output before seeing the output.

:show pronoun causes the system to display what the bindings of all pronouns are twice. Once after processing the user's input and once after processing the bot's response. Pronoun bindings are maintained by the system but only used if you make use of them in script.

:show input displays the things you send back into the system using the ^input function.

:show mark is not something you are likely to use but it displays in the log the propogation of marked items in your sentence. If you do :echo that stuff will also display on your screen.

:test ~topic

In my control scripts, I have a variable $testtopic, which if not null, will force the system to return to the named topic as the current one every volley. This allows me to use the :test command (which sets $testtopic to the given topic) to force the system to stay in a given topic for testing. An input may find an answer outside this topic (if there isn't a reply within the topic), but on next volley the system is back to the tested topic and if it's going to issue a gambit or check responders, it will check them from the tested topic first.

:testtopic ~topic sentence

This will execute responder mode of the named topic on the sentence given, to see what would happen. It forces focus on that topic, so no other topic could intervene. In addition to showing you the output it generates, it shows you the values of all user variables it changes.

:testpattern (…..) sentence

The system inputs the sentence and tests the pattern you provide against it. It tells you whether it matched or failed.

:testpattern (it died) Do you know if it died?

Some patterns require variables to be set up certain ways. You can perform assignments prior to the sentence.

:testpattern ($gender=male hit) $gender = male hit me

Typically you might use testpattern to see if a subset of your pattern that fails works, trying to identify what has gone wrong.

:nocomment

If you use :source and its lines have script comment data (# xxx) then using :nocomment 1 will tell ChatScript to ignore the comment on the input line.

:testoutput output script

Given an output script, the system executes it to show you what it would generate. It's a test scaffold for output script.

Data Display Commands

The system is filled with data, some of which you might want to see from time to time.

:variables

Rarely will the issue be that some variable of yours isn't correct. But you can show the values of all user variables using the :variables command.

:interesting

Show the interesting topics list.

:help

Displays available commands and a brief statement of purpose.

:systemvariables

Show the values of all system %variables.

:functions

Show all ^functions defined by the system.

:macros

Show all ^macros defined by the user.

:who

Show name of current user and current bot.

:word apple

Given a word, this displays the dictionary entry for it as well some data up it's hierarchy. The word is case sensitive and if you want to check out a composite word, you need to use underscores instead of blanks. So :word TV_star .

:sets apple

Hierarchically lists all the concepts (including topics) the word belongs to.

:facts

This prints out the current facts stored with the user.

:allfacts

This dumps a list of all the facts (including system facts) into the file TMP/facts.txt .

:dictstats

This displays a breakdown of how many of what kinds of things are in the dictionary.

:topics

This lists the names of all the topics defined in the system.

:up word

Shows the dictionary and concept hierarchy of things above the given word or concept.

:down word n

Shows the dictionary hierarchy below a word or, if the word is name of a concept, the members of the concept. Since displaying down can subsume a lot of entries, you can specify how many levels down to display (n). The default is 1.

:used topicname

This lists the top level rules that have been disabled for a topic, including from use or the ^disable function.

:findwords pattern
This uses the pattern of characters and * to name words and phrases in the dictionary matching it. E.g.
:findwords *_executive
   chief_executive
   railroad_executive
:findwords *f_exe*
   chief_executive
   chief_executive_officer
   Dancing_and_singing_are_my_idea_of_exercise.

:echo

Most commands send data to the log file AND temporarily turn on echoing at the console. For those that don't, if you want log data to show up on the console you can toggle :echo.

System Control Commands

:build

This compiles script into ready-to-use data in the TOPICS folder. You name a file to build. If the file name has a 0 at the end of it, it will build as level 0. Any other file name will build as level 1. You can build levels in any order or just update a single level.

A build file is named filesxxx.txt where the xxx part is what you specify to the build command. So :build angela0 will use filesAngela0.txt to build level 0. A build file has as its content a list of file paths to read as the script source data. It may also have comment lines starting with # . These paths are usually relative to the top level directory. E.g,

```
# ontology data
RAWDATA/ONTOLOGY/adverbhierarchy.top
RAWDATA/ONTOLOGY/adjectivehierarchy.top
RAWDATA/ONTOLOGY/prepositionhierarchy.top
```

Depending on what you put into it, a build file may build a single bot or lots of bots or a common set of data or whatever.

:bot angela

Change focus to conversing with the named bot (presuming you have such a bot).

:reset

Flush the current user's history, starting conversation from the beginning.

:switch username

Change your login id. It will then prompt you for new input as that user and proceed from there.

:all

This toggles a system mode where it will not stop after it first finds an output. It will find everything that matches and shows all the outputs. It just doesn't proceed to do gambits. Since it is showing everything, it erases nothing. There is a system variable %all you can query to see if the all mode is turned on. The boolean returns the null string or "1"

:repeat

This toggles a mode allowing the system to repeat anything it says. It's a sort of global temporary application of ^repeat() to all rules.

:noerase

This toggles a mode whereby the system may not erase any rule after being used, including gambits.

:execute function(args) – execute the function and args specified. Technically it executes the output data given, but there is little point in it being other than a function call. E.g., :execute pos(noun word plural) will output "words".

:source REGRESS/bigregress.txt

Switch the system to reading input from the named file. Good for regression testing. The system normally prints out just the output, while the log file contains both the input and the output. You can say *:source filename echo* to have input echoed to the console. If you say *:source filename internal* the system will echo the input, then echo the tokenized sentences it handled.

:testpos

This switches input to the named file (if not named defaults to REGRESS/posttest.txt) and running regression POS testing. If the result of processing an input deviates from that listed in the test file, the system presents this as an error.

:verifysub

This tests each substitution in the LIVEDATA/substitutes file to see if it does the expected thing.

:restart

This will force the system to reload all its data files from disk (dictionary, topic data, live data) and then ask for your login. It's like starting the system from scratch, but it never stops execution. Good for revising a live server.

:fakereply why
:fakereply ok

These commands cause the system to talk to itself. As the user it always says why or OK. Not something you are likely to use.

Debugging Function ^debug()

As a last ditch, you can add this function call into a pattern or the output and it will call DebugCode in functionExecute.cpp so you know exactly where you are and can use a debugger to follow code thereafter.

When all else fails

Usually you can email me for advice and solutions.